MASSACHVSETTS INSTITVTE OF TECHNOLOGY
6.946J, 8.351J, 12.620J

Classical Mechanics: A Computational Approach

Problem Set 7—Fall 2020

Issued: 14 October 2020 Due: 23 October 2020

Note: Project 3 is due on Friday, 30 October 2020; Project 4 is announced here

Reading: SICM2 Chapter 3 through section 3.7

Introduction

In the Hamiltonian formulation of dynamics the conserved quantities are especially valuable. We can simplify the analysis of a problem by reducing the problem to a system with one fewer degree of freedom for each conserved quantity. The solution of the reduced problem can be used to generate a solution of the full system using quadratures.

If there is not a conserved quantity for each degree of freedom the system exhibits the *divided phase space*: some initial conditions evolve regular trajectories and others evolve chaotic trajectories. The method of *surfaces of section*, invented by Poincaré and used to great advantage by Hénon and Heilas, allows one to visualize the structure of the phase space of a two degree-of-freedom system, so that the regular and chaotic trajectories can be discerned.

Exercises

- Exercise: An effective Lagrangian

  The axisymmetric top was attacked by consideration of three conserved quantities, the energy E, the angular momentum conjugate to the longitude p_ϕ, and the angular momentum conjugate to the rotation about the symmetry axis of the top p_ψ. This reduced the problem to the motion in a single degree of freedom, the tilt θ. The other degrees of freedom are then determined from their initial conditions by integration of a subsystem driven by the tilt. In Chapter 2 we used an "effective Lagrangian" to describe the θ motion. This strategy was justified in the Hamiltonian framework.

  The Lagrangian we used for the top is just

```
(define ((L-axisymmetric-top A C gMR) local)
  (let ((q (coordinate local)) (qdot (velocity local)))
    (let ((theta (ref q 0)) (thetadot (ref qdot 0))
          (phidot (ref qdot 1)) (psidot (ref qdot 2)))
      (+ (* 1/2 A
            (+ (square thetadot) (square (* phidot (sin theta)))))
         (* 1/2 C
            (square (+ psidot (* phidot (cos theta)))))
         (* -1 gMR (cos theta))))))
```

We can directly derive the effective Lagrangian using the Routhian procedure described in Section 3.4.1 "Lagrangian Reduction." Here is a short program that gets the effective Lagrangian for a Lagrangian. The program assumes a Lagrangian of the form $L(t; x, y; v_x, v_y)$, where x and y may have substructure. It performs a Legendre transform on v_y to get the Routhian.

```
(define ((Lagrangian->Routhian Lagrangian) R-state)
  (let ((t (time R-state))
        (q (coordinate R-state))
        (vp (ref R-state 2)))
    (let ((x (ref q 0))
          (y (ref q 1))
          (vx (ref vp 0))
          (py (ref vp 1)))
      (define (L vy)
        (Lagrangian (up t q (up vx vy))))
      ((Legendre-transform-procedure L) py))))
```

However, our Lagrangian state does not have this structure, so it is necessary to repackage the state to use this procedure.

1. Adjust the Lagrangian for the top to make $x = \theta$ and $y = (\phi, \psi)$. Compute the Routhian and compare it with the effective Lagrangian of Chapter 2.
2. Once we have a Routhian we can compute the equations of motion, which are a combination of Lagrange equations and Hamilton's equations. Here is a procedure that does this:

   ```
   (define (((Routh-equations Routhian) x y py) t)
     (define (L s)
       (let ((tau (time s))
             (q (coordinate s))
             (v (velocity s)))
         (Routhian (up tau (up q (y tau)) (up v (py tau))))))
     (define (H s)
       (let ((tau (time s))
             (q (coordinate s))
             (p (momentum s)))
         (Routhian (up tau (up (x tau) q) (up ((D x) tau) p)))))
     (up
      (((Lagrange-equations L) x) t)
      (((Hamilton-equations H) y py) t)))
   ```

   Show the Routh equations for the top. Observe that two of the momenta are automatically conserved.

- Exercise: The periodically-driven pendulum in the book is driven vertically. Consider an alternative drive where the pivot is moved horizontally. Formulate this system and explore surfaces of section for it. Are there interesting features?

Project 4 Announcement

Project 4 will be due on 13 November 2020. There will also be problem set 10 due on that date, but it will be shorter. Project 4 will be about structures in phase space.

- Exercise 4.9: Secondary Islands — SICM2 page 333